
Задание А
	Реализуйте на ассемблере программу, выполняющую следующие действия над набором целых чисел A[i], имеющих следующие параметры (см. Таблицу А), где i – порядковый номер числа в наборе, и операндами B и C; результат поместить в A[i]:
1) выполните логическую операцию «НЕ»: ¬A[i]
2) выполните логическую операцию «И»: A[i] & B
3) выполните логическую операцию «ИЛИ»: A[i] v B
4) выполните логическую операцию «сложение по модулю 2»: (A[i]  +  B)
5) выполните логическую операцию «стрелка Пирса»: A[i] ↑ B
6) выполните логическую операцию «штрих Шеффера»: A[i] / B
7) выполните логическую операцию «эквивалентность»: A[i] ≡ B
8) выполните операцию арифметического сложения: A[i] + B
9) выполните операцию арифметического вычитания: A[i] - B
10) выполните операцию арифметического умножения: A[i] · С
10) выполните операцию арифметического деления: A[i] / С
Рекомендуется соблюдать порядок выполнения указанных действий, обозначенных в конечном варианте задания. Все исходные числа представлены в дополнительном коде.

Таблица А – Параметры используемых переменных в рамках заданий типа «А»
	Набор A
	Операнд B
	Операнд C
	Требуемые действия

	Кол-во элементов
	Знаковое
	Размер числа, байт
	Расположение в памяти
	Адресация
	
	
	

	
	«1» – да
	1
	«1» – внутреннее ОЗУ
	«1» – прямая
	«0» – отсутствует
	

	
	«0» – нет
	2
	«2» – во внешнем ОЗУ*
	«2» – косвенная
	«1» – целое беззнаковое 2-байтовое число
	

	
	
	4
	
	
	«2» – целое беззнаковое 1-байтовое число
	

	
	
	
	
	
	«3» – целое знаковое 2-байтовое число
	

	
	
	
	
	
	«4» – целое знаковое    1-байтовое число
	

	* — выбранный вами контроллер должен иметь «встроенную» на кристалле микросхемы память класса XDATA, внешнюю по отношению к памяти ядра контроллера



	Код варианта А-6-0-2-1-2-1-0-[1, 3, 6]
	Расшифровка: задание типа «А» – массив A[i] с кол-вом элементов 6, числа беззнаковые двубайтные, расположены во внутреннем ОЗУ, для доступа к ним применяется косвенная адресация; операнд В – целое беззнаковое двухбайтное число; операнд С отсутствует; требуется выполнить лог. операции «НЕ», «ИЛИ» и «штрих Шеффера».





Текст программы с комментариями
DSEG AT 030h			; сегмент внутренней памяти данных с адреса 030h
	A_: DS 12			; резервируем место под набор из 6-ти
; 2-байтовых чисел

B_ EQU 9AE7h			; второй операнд (число B)

CSEG AT 000h			; сегмент кода с адреса 0
Start:
; инициализация набора чисел значениями
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV DPTR, #iniA		; загрузка адр. массива в памяти ком. в DPTR
	MOV R2, #12		; счетчик байтов
lp_ini:
	CLR A				; обнуляем аккумулятор
	MOVC A, @A+DPTR		; считываем байт из памяти программ в A
	MOV @R0, A			; записываем знач. A в ОЗУ по адресу в R0
	INC R0			; инкремент адреса в R0
	INC DPTR			; инкремент адреса в DPTR
	DJNZ R2, lp_ini		; цикл, пока R2 > 0

; выполнение логической операции "НЕ"
; A[i] = ¬A[i]
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #12		; счетчик байтов
lp_not:
	MOV A, @R0			; загрузка байта из последовательности
	CPL	A			; инверсия аккумулятора (лог. "НЕ")
	MOV @R0, A			; запись инвертированного значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_not		; цикл, пока R2 > 0

; выполнение логической операции "ИЛИ"
; A[i] = A[i] v B
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #6			; счетчик элементов
lp_or:
	MOV A, @R0			; загрузка ст. байта числа последовательности
	ORL	A, #HIGH(B_)	; лог. ИЛИ аккум-ра и ст. байта операнда B
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	MOV A, @R0			; загрузка мл. байта числа последовательности
	ORL	A, #LOW(B_)	; лог. ИЛИ аккум-ра и мл. байта операнда B
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_or		; цикл, пока R2 > 0

; выполнение логической операции "штрих Шеффера" (И-НЕ)
; A[i] = A[i] | B
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #6			; счетчик элементов
lp_nand:
	MOV A, @R0			; загрузка ст. байта числа последовательности
	ANL	A, #HIGH(B_)	; лог. И аккумулятора и ст. байта операнда B
	CPL A				; инверсия значения аккумулятора
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	MOV A, @R0			; загрузка мл. байта числа последовательности
	ANL	A, #LOW(B_)	; лог. И аккумулятора и мл. байта операнда B
	CPL A				; инверсия значения аккумулятора
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_nand		; цикл, пока R2 > 0

	SJMP Start			; зацикливание программы

; значения чисел набора
iniA:
	DW 3645h, 0F365h, 1234h, 0f705h, 0321h, 6754h
	
END



	Результаты выполнения
А) инициализация массива:
[image: ]





Б) выполнение лог. операции «НЕ»:
[image: ]

В) выполнение лог. операции «ИЛИ»:
[image: ]

Г) выполнение лог. операции «штрих Шеффера»
[image: ]

