
Задание Д
	Инициализируйте последовательный порт со следующими параметрами: … бит данных, … стоп-бит, скорость передачи BR = … бит/с (или бод/с) при установленной частоте тактирования МК – Fosc = ... . Напишите на ассемблере подпрограмму пересылки массива байтов через последовательный порт.
– Если в качестве задания по 2-ой части курса был указан вариант задачи типа «А», тогда модифицируйте свои программы таким образом, чтобы сформировать подпрограмму поочередной пересылки массива A[i] и значений переменных В и С через последователь-ный порт.
– Если в качестве задания по 2-ой части курса был указан вариант задачи типа «Б», тогда 
модифицируйте свои программы так, чтобы организовать подпрограмму пересылки значений переменных x, y, z, t, q (выбранный вариант задания может включать не все перечисленные переменные) и всех результатов через последовательный порт.
Возможности программы должны предусматривать вывод информации с заданным интервалом времени. Временной интервал следует формировать с помощью одного из имеющихся в составе МК таймеров-счетчиков (ТС0, ТС1, ТС2) или с помощью подпрограммы (П) или макрокоманды (М).
Протестируйте работу полученной программы.

	Код варианта Д-8-1-28900-4,9152-70-П
	Расшифровка: задание типа «Д»; кол-во бит передаваемых данных – 8; стоп-бит – присутствует; скорость передачи данных BR = 28900 бит/с; тактовая частота МК
Fosc = 4,9152 МГц; передача данных осуществляется с задержкой T = 70 мс; непосредственно временная задержка организуется с помощью подпрограммы.

Текст программы с комментариями
DSEG AT 030h			; сегмент внутренней памяти данных с адреса 030h
	A_: DS 12			; резервируем место под набор из 6-ти
					; 2-байтовых чисел

B_ EQU 9AE7h			; второй операнд (число B)

CSEG AT 000h			; сегмент кода с адреса 0
Start:
; инициализация набора чисел значениями
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV DPTR, #iniA		; загрузка адр. массива в памяти ком. в DPTR
	MOV R2, #12		; счетчик байтов
lp_ini:
	CLR A				; обнуляем аккумулятор
	MOVC A, @A+DPTR		; считываем байт из памяти программ в A
	MOV @R0, A			; записываем знач. A в ОЗУ по адресу в R0
	INC R0			; инкремент адреса в R0
	INC DPTR			; инкремент адреса в DPTR
	DJNZ R2, lp_ini		; цикл, пока R2 > 0

; выполнение логической операции "НЕ"
; A[i] = ¬A[i]
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #12		; счетчик байтов
lp_not:
	MOV A, @R0			; загрузка байта из последовательности
	CPL	A			; инверсия аккумулятора (лог. "НЕ")
	MOV @R0, A			; запись инвертированного значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_not		; цикл, пока R2 > 0

; выполнение логической операции "ИЛИ"
; A[i] = A[i] v B
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #6			; счетчик элементов
lp_or:
	MOV A, @R0			; загрузка ст. байта числа последовательности
	ORL	A, #HIGH(B_)	; лог. ИЛИ аккум-ра и ст. байта операнда B
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	MOV A, @R0			; загрузка мл. байта числа последовательности
	ORL	A, #LOW(B_)	; лог. ИЛИ аккум-ра и мл. байта операнда B
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_or		; цикл, пока R2 > 0

; выполнение логической операции "штрих Шеффера" (И-НЕ)
; A[i] = A[i] | B
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #6			; счетчик элементов
lp_nand:
	MOV A, @R0			; загрузка ст. байта числа последовательности
	ANL	A, #HIGH(B_)	; лог. И аккумулятора и ст. байта операнда B
	CPL A				; инверсия значения аккумулятора
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	MOV A, @R0			; загрузка мл. байта числа последовательности
	ANL	A, #LOW(B_)	; лог. И аккумулятора и мл. байта операнда B
	CPL A				; инверсия значения аккумулятора
	MOV @R0, A			; запись значения в память
	INC R0			; инкремент адреса в R0
	DJNZ R2, lp_nand	; цикл, пока R2 > 0

	; запись B после массива
	MOV @R0, #HIGH(B_)	; ст. байт
	INC R0			; инкремент адреса в R0
	MOV @R0, #LOW(B_)	; мл. байт

	; настройка последовательного порта
	ORL PCON, #80h		; SMOD = 1 (удвоенная скорость)
	MOV SCON, #50h		; установка 1-го режима работы UART
	; настройка таймера TC1 для 1-го режима работы UART
	ANL TMOD, #0Fh 	
	ORL TMOD, #20h		; выбор режима 2 работы TC1
	MOV TH1, #0xFF		; значение для скорости BR = 25600 (28900) бит/с 
	SETB TR1 			; запустить TC1

	; передача через UART набора A
	MOV R0, #A_		; загрузка адреса массива в регистр R0
	MOV R2, #6+1		; счетчик чисел в наборе + операнд B
	CALL sArr			; вызов подпрограммы передачи массива

	CLR TR1			; остановить ТС1
	SJMP Start			; зацикливание программы

; значения чисел набора
iniA:
	DW 3645h, 0F365h, 1234h, 0f705h, 0321h, 6754h

; подпрограмма временной задержки в 70 мс
p70:
	mov R3, #56		; загрузка 
	mov R4, #197		; счетчиков циклов
lp_p:
	djnz R4, lp_p		; циклы
	djnz R3, lp_p		; задержки
	RET				; возврат из подпрограммы

; подпрограмма передачи массива через последовательный порт
; R0 - адрес массива
; R2 - кол-во передаваемых чисел
sArr:
	CLR TI			; сброс флага завершения передачи
	MOV SBUF, @R0		; передача ст. байта числа
	JNB TI, $			; ожидание завершения передачи
	INC R0			; инкремент адреса в R0
	CLR TI			; сброс флага завершения передачи
	MOV SBUF, @R0		; передача мл. байта числа
	JNB TI, $			; ожидание завершения передачи
	INC R0			; инкремент адреса в R0
	CALL p70			; вызов подпрограммы задержки 70 мс
	DJNZ R2, sArr		; цикл, пока R2 > 0
	CLR TI			; сброс флага завершения передачи
	RET				; возврат из подпрограммы

END

	Результаты выполнения
А) задержка после передачи в 70 мс:
[image: ]

Б) передача всех чисел массива и операнда B:
[image: ]
[bookmark: _GoBack]

